# This file is part of BlackArch Linux ( https://www.blackarch.org/ ).
# See COPYING for license details.

pkgname=python-ptable-boxchar
pkgver=243.eb76080
pkgrel=4
pkgdesc='Simple Python library designed to make it quick and easy to represent tabular data in visually appealing ASCII tables.'
arch=('any')
url='https://github.com/platomav/PTable'
license=('BSD')
depends=('python')
makedepends=('git' 'python-setuptools')
conflicts=('python-ptable')
source=('git+https://github.com/platomav/PTable.git')
sha512sums=('SKIP')
_dir_name_=PTable

pkgver() {
  cd "$_dir_name_"

  echo $(git rev-list --count HEAD).$(git rev-parse --short HEAD)
}

build() {
  cd "$_dir_name_"

  python setup.py build
}

package() {
  cd "$_dir_name_"

  python setup.py install --prefix=/usr --root="$pkgdir" --optimize=1 \
    --skip-build
}

